

Description

`estimates selected` reports on coefficients from one or more estimation results. It creates a table that indicates which coefficients were estimated in each model and, if requested, reports the value of those coefficients. The results may be sorted based on the values of the estimated coefficients or based on variable names.

Quick start

Compare coefficients for stored estimates `m1` and `m2`

```
estimates selected m1 m2
```

Same as above, but display a `u` for covariates that are not specified in the model

```
estimates selected m1 m2, display(u)
```

Compare stored estimates `l1` and `l2`, and order the rows by absolute values of the coefficients

```
estimates selected l1 l2, sort(coef)
```

Menu

Statistics > Postestimation

Syntax

```
estimates selected [namelist] [ , options ]
```

namelist is the name given to previously stored estimation results, a list of names, `_all`, or `*`. A name may be `.`, meaning the current (active) estimates. `_all` and `*` mean the same thing.

<i>options</i>	Description
Main	
<code>display(<i>info</i>)</code>	display <i>info</i> ; default is <code>display(x)</code>
<code>sort(<i>on</i>)</code>	sort rows in order of <i>on</i>
Reporting	
<code>noabbrev</code>	do not abbreviate variable names
<code>display_options</code>	control row spacing, line width, and display of omitted variables and base and empty cells

`collect` is allowed; see [U] 11.1.10 Prefix commands.

Options

Main

`display(info)` specifies what to display in the table. The default is `display(x)`.

Blank cells in the table indicate that the corresponding covariate does not have a fitted value. For some covariates without fitted values, a code that indicates the reason for omission is reported in the table. Base levels of factors and interactions are coded with the letter b. Empty levels of factors and interactions are coded with the letter e. Covariates omitted because of collinearity are coded with the letter o.

`display(x)` displays an x in the cell of the table where a covariate has a fitted value. This is the default.

`display(u)` is the same as `display(x)`, except that when a covariate was not specified in the model, u (for unavailable) is displayed instead of a blank cell.

`display(coef [, eform fformat(%fmt)])` specifies that coefficient values be displayed in the table.

`eform` displays coefficients in exponentiated form. For each coefficient, e^b rather than b is displayed. This option can be used to display odds ratios, incidence-rate ratios, relative-risk ratios, hazard ratios, and subhazard ratios after the appropriate estimation command.

`format(%fmt)` specifies the display format for the coefficients in the table. The default is `format(%9.0g)`.

`sort(on)` specifies how to sort the rows of the table. By default, coefficients are displayed in the order in which they appear in the estimation results.

`sort(none)` specifies that the rows are not sorted. This is the default. The order of the coefficients is taken from their order in `e(b)`.

`sort(names)` orders rows alphabetically by the variable names of the covariates. In the case of factor variables, main effects and nonfactor variables are displayed first in alphabetical order; then all two-way interactions are displayed in alphabetical order, then all three-way interactions, and so on.

`sort(coef)` orders rows in descending order by the absolute values of the coefficients. When results from two or more estimation results are displayed, results are sorted first by the ordering for the first estimation result with rows representing coefficients not in the first estimation result last. Within the rows representing coefficients not in the first estimation result, the rows are sorted by the ordering for the second estimation result with rows representing coefficients not in the first or second estimation results last. And so on.

Reporting

`noabbrev` prevents variable names from being abbreviated in the row titles of the table. Long variable names are split onto multiple lines if they do not fit.

`display_options:` `vsquish`, `fvwrap(#)`, `fvwrapon(style)`, and `no!stretch`; see [\[R\] Estimation options](#).

Remarks and examples

`estimates selected` produces a table based on estimated coefficients from one or more models. Results can be sorted by the values of the estimated coefficients or by variable names. Multiple models are displayed side-by-side, making it easy to compare which covariates were included in each model or to compare the estimated values of those coefficients.

► Example 1: Compare coefficients across models

To compare coefficients from two or more models, we first need to store the results of each model using `estimates store`; see [\[R\] estimates store](#). Then we use `estimates selected` to obtain a table comparing which coefficients were estimated in each model. By default, the table reports an x for each coefficient that was estimated.

```
. use https://www.stata-press.com/data/r19/auto
(1978 automobile data)

. regress mpg gear turn
(output omitted)

. estimates store small

. regress mpg gear turn length
(output omitted)

. estimates store large

. estimates selected small large
```

	small	large
gear_ratio	x	x
turn	x	x
length		x
_cons	x	x

Legend:

b - base level
e - empty cell
o - omitted
x - estimated

There is only one difference in the two `regress` commands; the `large` model includes `length` as a covariate, but the `small` model does not. Therefore, the above table displays x's for all but the `length` coefficient in the `small` model. When working with larger models or with more models, this type of table makes it easy to spot the differences in the model specification.

By default, the rows of the table in the order that covariates appear in the models. To sort the results on the covariate names, we add the `sort(names)` option.

```
. estimates selected small large, sort(names)
```

	small	large
gear_ratio	x	x
length		x
turn	x	x
_cons	x	x

Legend:

b - base level
e - empty cell
o - omitted
x - estimated

We could have instead sorted on the values of the estimated coefficients by including the `sort(coef)` option.

To display the coefficients values instead of the x's, we add the `display(coef)` option.

```
. estimates selected small large, sort(names) display(coef)
```

	small	large
gear_ratio	3.032884	1.35666
length		-.1665899
turn	-.7330502	-.1219185
_cons	41.21801	53.3487

Legend:

b - base level
e - empty cell
o - omitted



Stored results

`estimates selected` stores the following in `r()`:

Macros

`r(names)` names of results used

Matrices

`r(coef)` matrix M : $n \times m$
 $M[i, j] = i$ th coefficient estimate for model j ; $i = 1, \dots, n$; $j = 1, \dots, m$

Also see

[\[R\] estimates](#) — Save and manipulate estimation results

[\[LASSO\] lassocoeff](#) — Display coefficients after lasso estimation results

Stata, Stata Press, and Mata are registered trademarks of StataCorp LLC. Stata and Stata Press are registered trademarks with the World Intellectual Property Organization of the United Nations. StataNow and NetCourseNow are trademarks of StataCorp LLC. Other brand and product names are registered trademarks or trademarks of their respective companies. Copyright © 1985–2025 StataCorp LLC, College Station, TX, USA. All rights reserved.



For suggested citations, see the FAQ on [citing Stata documentation](#).